Rules for integrands of the form $u\,(a+b\,\text{ArcCosh}[c\,x])^n$

1. $\int (d+e\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$

   1. $\int (d+e\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

      1: $\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{d+e\,x}\,dx$

Derivation: Integration by substitution

Basis: $\frac{1}{d+e\,x} = \text{Subst}\left[\frac{\text{Sinh}[x]}{c\,d+e\,\text{Cosh}[x]},\,x,\,\text{ArcCosh}[c\,x]\right]\,\partial_x\text{ArcCosh}[c\,x]$

Note: $\frac{(a+b\,x)^n\,\text{Sinh}[x]}{c\,d+e\,\text{Cosh}[x]}$ is not integrable unless $n\in\mathbb{Z}^+$.

Rule: If $n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{d+e\,x}\,dx \longrightarrow \text{Subst}\left[\int \frac{(a+b\,x)^n\,\text{Sinh}[x]}{c\,d+e\,\text{Cosh}[x]}\,dx,\,x,\,\text{ArcCosh}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(d_.+e_.*x_),x_Symbol] :=
  Subst[Int[(a+b*x)^n*Sinh[x]/(c*d+e*Cosh[x]),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[n,0]
```

2: $\int (d+e x)^m (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \land m \neq -1$

Reference: G&R 2.832, CRC 454, A&S 4.4.67

Derivation: Integration by parts

Basis: $(d+e x)^m == \partial_x \frac{(d+e x)^{m+1}}{e (m+1)}$

Basis: $\partial_x (a+b \operatorname{ArcCosh}[c x])^n == \frac{b c n (a+b \operatorname{ArcCosh}[c x])^{n-1}}{\sqrt{-1+c x} \sqrt{1+c x}}$

Rule: If $n \in \mathbb{Z}^+ \land m \neq -1$, then

$$\int (d+e x)^m (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow \frac{(d+e x)^{m+1} (a+b \operatorname{ArcCosh}[c x])^n}{e (m+1)} - \frac{b c n}{e (m+1)} \int \frac{(d+e x)^{m+1} (a+b \operatorname{ArcCosh}[c x])^{n-1}}{\sqrt{-1+c x} \sqrt{1+c x}} \, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*ArcCosh[c*x])^n/(e*(m+1)) -
  b*c*n/(e*(m+1))*Int[(d+e*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

2. $\int (d+e x)^m (a+b\,\text{ArcCosh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+$

1: $\int (d+e x)^m (a+b\,\text{ArcCosh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+ \land n < -1$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}^+ \land n < -1$, then

$$\int (d+e x)^m (a+b\,\text{ArcCosh}[c x])^n \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(d+e x)^m (a+b\,\text{ArcCosh}[c x])^n, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[m,0] && LtQ[n,-1]
```

2: $\int (d+e x)^m (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] = \frac{1}{c} \operatorname{Subst}\left[\operatorname{Sinh}[x] \, F\left[\frac{\operatorname{Cosh}[x]}{c}\right], x, \operatorname{ArcCosh}[c x]\right] \partial_x \operatorname{ArcCosh}[c x]$

Basis: If $m \in \mathbb{Z}$, then $(d+e x)^m = \frac{1}{c^{m+1}} \operatorname{Subst}\left[\operatorname{Sinh}[x] \left(c d+e \operatorname{Cosh}[x]\right)^m, x, \operatorname{ArcCosh}[c x]\right] \partial_x \operatorname{ArcCosh}[c x]$

Note: If $m \in \mathbb{Z}^+$, then $(a+b x)^n \operatorname{Sinh}[x] (c d+e \operatorname{Cosh}[x])^m$ is integrable in closed-form.

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (d+e x)^m (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow \frac{1}{c^{m+1}} \operatorname{Subst}\left[\int (a+b x)^n \operatorname{Sinh}[x] (c d+e \operatorname{Cosh}[x])^m \, dx, x, \operatorname{ArcCosh}[c x]\right]$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Sinh[x]*(c*d+e*Cosh[x])^m,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[m,0]
```

2. $\int P_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$

1: $\int P_x\,(a+b\,\text{ArcCosh}[c\,x])\,dx$

– Derivation: Integration by parts and piecewise constant extraction

– Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x]) == \frac{b\,c}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}$

– Basis: $\partial_x\,\frac{\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}} == 0$

– Rule: Let $u\to\int P_x\,dx$, then

$$\int P_x\,(a+b\,\text{ArcCosh}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcCosh}[c\,x])-b\,c\int\frac{u}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\,dx$$

$$\longrightarrow u\,(a+b\,\text{ArcCosh}[c\,x])-\frac{b\,c\,\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\int\frac{u}{\sqrt{1-c^2\,x^2}}\,dx$$

– Program code:

```
Int[Px_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[ExpandExpression[Px,x],x]},
  Dist[a+b*ArcCosh[c*x],u,x] -
  b*c*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Px,x]
```

x: $\int P_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x])^n == \frac{b\,c\,n\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}$

Basis: $\partial_x\,\frac{\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}} == 0$

Rule: If $n\in\mathbb{Z}^+$, let $u\to\int P_x\,dx$, then

$$\int P_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\;\longrightarrow\;u\,(a+b\,\text{ArcCosh}[c\,x])^n-b\,c\,n\int\frac{u\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\,dx$$

$$\longrightarrow\;u\,(a+b\,\text{ArcCosh}[c\,x])^n-\frac{b\,c\,n\,\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\int\frac{u\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
(* Int[Px_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  With[{u=IntHide[Px,x]},
  Dist[(a+b*ArcCosh[c*x])^n,u,x] -
  b*c*n*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[u*(a+b*ArcCosh[c*x])^(n-1)/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c},x] && PolyQ[Px,x] && IGtQ[n,0] *)
```

2: $\int P_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\neq 1$

Derivation: Algebraic expansion

Rule: If $n\neq 1$, then

$$\int P_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[P_x\,(a+b\,\text{ArcCosh}[c\,x])^n,\,x\right]dx$$

Program code:

```
Int[Px_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[Px*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,n},x] && PolyQ[Px,x]
```

3. $\int P_x\,(d+e\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

1: $\int P_x\,(d+e\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])\,dx$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x]) == \frac{b\,c}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}$

Basis: $\partial_x\,\frac{\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}} == 0$

Rule: Let $u\to\int P_x\,(d+e\,x)^m\,dx$, then

$$\int P_x\,(d+e\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])\,dx\;\longrightarrow\;u\,(a+b\,\text{ArcCosh}[c\,x])-b\,c\int\frac{u}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\,dx$$

$$\longrightarrow u\,(a + b\,\text{ArcCosh}[c\,x]) - \frac{b\,c\,\sqrt{1 - c^2 x^2}}{\sqrt{-1 + c\,x}\,\sqrt{1 + c\,x}} \int \frac{u}{\sqrt{1 - c^2 x^2}}\,dx$$

– Program code:

```
Int[Px_*(d_.+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[Px*(d+e*x)^m,x]},
  Dist[a+b*ArcCosh[c*x],u,x] -
  b*c*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,m},x] && PolyQ[Px,x]
```

2: $\int (f + g\,x)^p\,(d + e\,x)^m\,(a + b\,\text{ArcCosh}[c\,x])^n\,dx$ when $(n \mid p) \in \mathbb{Z}^+ \land m \in \mathbb{Z}^- \land m + p + 1 < 0$

– Derivation: Integration by parts

– Basis: $\partial_x\,(a + b\,\text{ArcCosh}[c\,x])^n == \frac{b\,c\,n\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}$

– Note: If $p \in \mathbb{Z}^+ \land m \in \mathbb{Z}^- \land m + p + 1 < 0$, then $\int (f + g\,x)^p\,(d + e\,x)^m\,dx$ is a rational function.

– Rule: If $(n \mid p) \in \mathbb{Z}^+ \land m \in \mathbb{Z}^- \land m + p + 1 < 0$, let $u \to \int (f + g\,x)^p\,(d + e\,x)^m\,dx$, then

$$\int (f + g\,x)^p\,(d + e\,x)^m\,(a + b\,\text{ArcCosh}[c\,x])^n\,dx \longrightarrow u\,(a + b\,\text{ArcCosh}[c\,x])^n - b\,c\,n \int \frac{u\,(a + b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{-1 + c\,x}\,\sqrt{1 + c\,x}}\,dx$$

– Program code:

```
Int[(f_.+g_.*x_)^p_.*(d_+e_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  With[{u=IntHide[(f+g*x)^p*(d+e*x)^m,x]},
  Dist[(a+b*ArcCosh[c*x])^n,u,x] -
  b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[m,0] && LtQ[m+p+1,0]
```

3: $\int \frac{(f+g x+h x^2)^p (a+b \text{ArcCosh}[c x])^n}{(d+e x)^2} dx$ when $(n \mid p) \in \mathbb{Z}^+ \land e g-2 d h==0$

- Derivation: Integration by parts

- Basis: $\partial_x (a+b \text{ArcCosh}[c x])^n == \frac{b c n (a+b \text{ArcCosh}[c x])^{n-1}}{\sqrt{-1+c x} \sqrt{1+c x}}$

  Note: If $p \in \mathbb{Z}^+ \land e g-2 d h == 0$, then $\int \frac{(f+g x+h x^2)^p}{(d+e x)^2} dx$ is a rational function.

- Rule: If $(n \mid p) \in \mathbb{Z}^+ \land e g-2 d h == 0$, let $u \to \int \frac{(f+g x+h x^2)^p}{(d+e x)^2} dx$, then

$$\int \frac{(f+g x+h x^2)^p (a+b \text{ArcCosh}[c x])^n}{(d+e x)^2} dx \longrightarrow u (a+b \text{ArcCosh}[c x])^n - b c n \int \frac{u (a+b \text{ArcCosh}[c x])^{n-1}}{\sqrt{-1+c x} \sqrt{1+c x}} dx$$

- Program code:

```
Int[(f_.+g_.*x_+h_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/(d_+e_.*x_)^2,x_Symbol] :=
  With[{u=IntHide[(f+g*x+h*x^2)^p/(d+e*x)^2,x]},
  Dist[(a+b*ArcCosh[c*x])^n,u,x] -
  b*c*n*Int[SimplifyIntegrand[u*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[n,0] && IGtQ[p,0] && EqQ[e*g-2*d*h,0]
```

4: $\int P_x (d + e x)^m (a + b \text{ArcCosh}[c x])^n \, dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, then

$$\int P_x (d + e x)^m (a + b \text{ArcCosh}[c x])^n \, dx \longrightarrow \int \text{ExpandIntegrand}\left[P_x (d + e x)^m (a + b \text{ArcCosh}[c x])^n, x\right] dx$$

Program code:

```
Int[Px_*(d_+e_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(d+e*x)^m*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x] && IGtQ[n,0] && IntegerQ[m]
```

4. $\int F[f+g x]\,(d+e x^2)^p\,(a+b\,\text{ArcCosh}[c x])^n\,dx$ when $c^2 d+e=0 \land p-\frac{1}{2}\in\mathbb{Z}$

1: $\int (f+g x)^m\,(d+e x^2)^p\,(a+b\,\text{ArcCosh}[c x])^n\,dx$ when $c^2 d+e=0 \land p-\frac{1}{2}\in\mathbb{Z} \land m\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c^2 d+e=0$, then $\partial_x \frac{(d+e x^2)^p}{(-1+c x)^p\,(1+c x)^p}=0$

– Rule: If $c^2 d+e=0 \land p-\frac{1}{2}\in\mathbb{Z} \land m\in\mathbb{Z}$, then

$$\int (f+g x)^m\,(d+e x^2)^p\,(a+b\,\text{ArcCosh}[c x])^n\,dx \longrightarrow$$
$$\frac{(-d)^{\text{IntPart}[p]}\,(d+e x^2)^{\text{FracPart}[p]}}{(-1+c x)^{\text{FracPart}[p]}\,(1+c x)^{\text{FracPart}[p]}}\int (f+g x)^m\,(-1+c x)^p\,(1+c x)^p\,(a+b\,\text{ArcCosh}[c x])^n\,dx$$

– Program code:

```
Int[(f_+g_.*x_)^m_.*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^IntPart[p]*(d+e*x^2)^FracPart[p]/((-1+c*x)^FracPart[p]*(1+c*x)^FracPart[p])*
    Int[(f+g*x)^m*(-1+c*x)^p*(1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2] && IntegerQ[m]
```

2: $\int \text{Log}[h\,(f+g x)^m]\,(d+e x^2)^p\,(a+b\,\text{ArcCosh}[c x])^n\,dx$ when $c^2 d+e=0 \land p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c^2 d+e=0$, then $\partial_x \frac{(d+e x^2)^p}{(-1+c x)^p\,(1+c x)^p}=0$

– Rule: If $c^2 d+e=0 \land p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int \text{Log}[h\,(f+g x)^m]\,(d+e x^2)^p\,(a+b\,\text{ArcCosh}[c x])^n\,dx \longrightarrow$$

$$\frac{(-d)^{\text{IntPart}[p]} \left(d+e x^2\right)^{\text{FracPart}[p]}}{(-1+c x)^{\text{FracPart}[p]} (1+c x)^{\text{FracPart}[p]}} \int \text{Log}\left[h (f+g x)^m\right] (-1+c x)^p (1+c x)^p (a+b \text{ArcCosh}[c x])^n \, dx$$

Program code:

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d)^IntPart[p]*(d+e*x^2)^FracPart[p]/((-1+c*x)^FracPart[p]*(1+c*x)^FracPart[p])*
    Int[Log[h*(f+g*x)^m]*(-1+c*x)^p*(1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2]
```

5. $\int F[f+g x]\ (d1+e1\ x)^p\ (d2+e2\ x)^p\ (a+b\ \text{ArcCosh}[c\ x])^n\ dx$ when $e1 == c\ d1\ \land\ e2 == -c\ d2\ \land\ p-\frac{1}{2}\in\mathbb{Z}$

1. $\int (f+g\ x)^m\ (d1+e1\ x)^p\ (d2+e2\ x)^p\ (a+b\ \text{ArcCosh}[c\ x])^n\ dx$ when $e1 == c\ d1\ \land\ e2 == -c\ d2\ \land\ m\in\mathbb{Z}\ \land\ p-\frac{1}{2}\in\mathbb{Z}$

1. $\int (f+g\ x)^m\ (d1+e1\ x)^p\ (d2+e2\ x)^p\ (a+b\ \text{ArcCosh}[c\ x])^n\ dx$ when $e1 == c\ d1\ \land\ e2 == -c\ d2\ \land\ m\in\mathbb{Z}\ \land\ p-\frac{1}{2}\in\mathbb{Z}\ \land\ d1>0\ \land\ d2<0$

1: $\int (f+g\ x)^m\ (d1+e1\ x)^p\ (d2+e2\ x)^p\ (a+b\ \text{ArcCosh}[c\ x])\ dx$ when $e1 == c\ d1\ \land\ e2 == -c\ d2\ \land\ m\in\mathbb{Z}^+\ \land\ p+\frac{1}{2}\in\mathbb{Z}^-\ \land\ d1>0\ \land\ d2<0\ \land\ (m>3\ \lor\ m<-2p-1)$

Derivation: Integration by parts

- Basis: $\partial_x\ (a+b\ \text{ArcCosh}[c\ x]) == \frac{b\ c}{\sqrt{-1+c\ x}\ \sqrt{1+c\ x}}$

Note: If $m\in\mathbb{Z}\ \land\ p+\frac{1}{2}\in\mathbb{Z}\ \land\ 0<m<-2p-1$, then $\int (f+g\ x)^m\ (d1+e1\ x)^p\ (d2+e2\ x)^p\ dx$ is an algebraic function.

- Rule: If $e1 == c\ d1\ \land\ e2 == -c\ d2\ \land\ m\in\mathbb{Z}^+\ \land\ p+\frac{1}{2}\in\mathbb{Z}^-\ \land\ d1>0\ \land\ d2<0\ \land\ (m>3\ \lor\ m<-2p-1)$, let $u\to\int (f+g\ x)^m\ (d1+e1\ x)^p\ (d2+e2\ x)^p\ dx$, then

$$\int (f+g\ x)^m\ (d1+e1\ x)^p\ (d2+e2\ x)^p\ (a+b\ \text{ArcCosh}[c\ x])\ dx \longrightarrow u\ (a+b\ \text{ArcCosh}[c\ x]) - b\ c\int \frac{u}{\sqrt{-1+c\ x}\ \sqrt{1+c\ x}}\ dx$$

- Program code:

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f+g*x)^m*(d1+e1*x)^p*(d2+e2*x)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[Dist[1/(Sqrt[-1+c*x]*Sqrt[1+c*x]),u,x],x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[m,0] && ILtQ[p+1/2,0] && GtQ[d1,0] && LtQ[d2,0] &&
  (GtQ[m,3] || LtQ[m,-2*p-1])
```

2: $\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^+ \land p+\frac{1}{2} \in \mathbb{Z} \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^+ \land p+\frac{1}{2} \in \mathbb{Z} \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$, then

$$\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow \int (d1+e1 x)^p (d2+e2 x)^p (a+b \text{ArcCosh}[c x])^n \, \text{ExpandIntegrand}\left[(f+g x)^m, x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[m,0] && IntegerQ[p+1/2] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0] &&
  (EqQ[n,1] && GtQ[p,-1] || GtQ[p,0] || EqQ[m,1] || EqQ[m,2] && LtQ[p,-2])
```

3. $\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land p+\frac{1}{2} \in \mathbb{Z}^+ \land d1>0 \land d2<0$

1: $\int (f+g x)^m \sqrt{d1+e1 x}\, \sqrt{d2+e2 x}\, (a+b \text{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^- \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$

- Derivation: Integration by parts

- Basis: If $e1 == c\, d1 \land e2 == -c\, d2 \land d1>0 \land d2<0$, then $\frac{(a+b \text{ArcCosh}[c x])^n}{\sqrt{d1+e1 x}\, \sqrt{d2+e2 x}} == \partial_x \frac{(a+b \text{ArcCosh}[c x])^{n+1}}{b c \sqrt{-d1\, d2}\, (n+1)}$

- Rule: If $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^- \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$, then

$$\int (f+g x)^m \sqrt{d1+e1 x}\, \sqrt{d2+e2 x}\, (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f+g x)^m \left(d1\, d2+e1\, e2\, x^2\right) (a+b \text{ArcCosh}[c x])^{n+1}}{b c \sqrt{-d1\, d2}\, (n+1)} -$$

$$\frac{1}{b c \sqrt{-d1\, d2}\, (n+1)} \int \left(d1\, d2\, g\, m+2\, e1\, e2\, f\, x+e1\, e2\, g\, (m+2)\, x^2\right) (f+g x)^{m-1} (a+b \text{ArcCosh}[c x])^{n+1} \, dx$$

- Program code:

```
Int[(f_+g_.*x_)^m_*Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d1*d2+e1*e2*x^2)*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  1/(b*c*Sqrt[-d1*d2]*(n+1))*Int[(d1*d2*g*m+2*e1*e2*f*x+e1*e2*g*(m+2)*x^2)*(f+g*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && ILtQ[m,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0]
```

2: $\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land p+\frac{1}{2} \in \mathbb{Z}^+ \land d1 > 0 \land d2 < 0 \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land p+\frac{1}{2} \in \mathbb{Z}^+ \land d1 > 0 \land d2 < 0 \land n \in \mathbb{Z}^+$, then

$$\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow \int \sqrt{d1+e1 x} \sqrt{d2+e2 x} (a+b \text{ArcCosh}[c x])^n \text{ExpandIntegrand}\left[(f+g x)^m \left(d+e x^2\right)^{p-1/2}, x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n,(f+g*x)^m*(d1+e1*x)^(p-1/2)*(d2+e2*x)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && IGtQ[p+1/2,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0]
```

3: $\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^- \land p-\frac{1}{2} \in \mathbb{Z}^+ \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$

- Derivation: Integration by parts

- Basis: If $e1 == c\, d1 \land e2 == -c\, d2 \land d1>0 \land d2<0$, then $\frac{(a+b \operatorname{ArcCosh}[c x])^n}{\sqrt{d1+e1 x}\,\sqrt{d2+e2 x}} == \partial_x \frac{(a+b \operatorname{ArcCosh}[c x])^{n+1}}{b c \sqrt{-d1\, d2}\,(n+1)}$

- Rule: If $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^- \land p-\frac{1}{2} \in \mathbb{Z}^+ \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$, then

$$\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f+g x)^m (d1+e1 x)^{p+\frac{1}{2}} (d2+e2 x)^{p+\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n+1}}{b c \sqrt{-d1\, d2}\,(n+1)} - \frac{1}{b c \sqrt{-d1\, d2}\,(n+1)} \cdot$$

$$\int (f+g x)^{m-1} (a+b \operatorname{ArcCosh}[c x])^{n+1} \operatorname{ExpandIntegrand}\left[\left(d1\, d2\, g\, m + e1\, e2\, f\, (2 p+1)\, x + e1\, e2\, g\, (m+2 p+1)\, x^2\right) (d1+e1 x)^{p-\frac{1}{2}} (d2+e2 x)^{p-\frac{1}{2}}, x\right] dx$$

- Program code:

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f+g*x)^m*(d1+e1*x)^(p+1/2)*(d2+e2*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  1/(b*c*Sqrt[-d1*d2]*(n+1))*
    Int[ExpandIntegrand[(f+g*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),
      (d1*d2*g*m+e1*e2*f*(2*p+1)*x+e1*e2*g*(m+2*p+1)*x^2)*(d1+e1*x)^(p-1/2)*(d2+e2*x)^(p-1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && ILtQ[m,0] && IGtQ[p-1/2,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0]
```

4. $\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land p-\frac{1}{2} \in \mathbb{Z}^- \land d1>0 \land d2<0$

1. $\int \frac{(f+g x)^m (a+b \text{ArcCosh}[c x])^n}{\sqrt{d1+e1 x} \sqrt{d2+e2 x}} \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land d1>0 \land d2<0$

1: $\int \frac{(f+g x)^m (a+b \text{ArcCosh}[c x])^n}{\sqrt{d1+e1 x} \sqrt{d2+e2 x}} \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^+ \land d1>0 \land d2<0 \land n<-1$

Derivation: Integration by parts

– Basis: If $e1 == c\, d1 \land e2 == -c\, d2 \land d1>0 \land d2<0$, then $\frac{(a+b \text{ArcCosh}[c x])^n}{\sqrt{d1+e1 x} \sqrt{d2+e2 x}} == \partial_x \frac{(a+b \text{ArcCosh}[c x])^{n+1}}{b c \sqrt{-d1\, d2}\, (n+1)}$

– Rule: If $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z}^+ \land d1>0 \land d2<0 \land n<-1$, then

$$\int \frac{(f+g x)^m (a+b \text{ArcCosh}[c x])^n}{\sqrt{d1+e1 x} \sqrt{d2+e2 x}} \, dx \to \frac{(f+g x)^m (a+b \text{ArcCosh}[c x])^{n+1}}{b c \sqrt{-d1\, d2}\, (n+1)} - \frac{g m}{b c \sqrt{-d1\, d2}\, (n+1)} \int (f+g x)^{m-1} (a+b \text{ArcCosh}[c x])^{n+1} \, dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  (f+g*x)^m*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  g*m/(b*c*Sqrt[-d1*d2]*(n+1))*Int[(f+g*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[m,0] && GtQ[d1,0] && LtQ[d2,0] && LtQ[n,-1]
```

2: $\int \frac{(f+g x)^m (a+b \operatorname{ArcCosh}[c x])^n}{\sqrt{d1+e1 x} \sqrt{d2+e2 x}} dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land d1 > 0 \land d2 < 0 \land (m > 0 \lor n \in \mathbb{Z}^+)$

Derivation: Integration by substitution

Basis: If $e1 == c\, d1 \land e2 == -c\, d2 \land d1 > 0 \land d2 < 0$, then
$\frac{F[x]}{\sqrt{d1+e1 x} \sqrt{d2+e2 x}} == \frac{1}{c \sqrt{-d1\, d2}} \operatorname{Subst}\left[F\left[\frac{\operatorname{Cosh}[x]}{c}\right], x, \operatorname{ArcCosh}[c x]\right] \partial_x \operatorname{ArcCosh}[c x]$

Note: *Mathematica* 8 is unable to validate antiderivatives of ArcCosh rule when c is symbolic.

Rule: If $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land d1 > 0 \land d2 < 0 \land (m > 0 \lor n \in \mathbb{Z}^+)$, then

$$\int \frac{(f+g x)^m (a+b \operatorname{ArcCosh}[c x])^n}{\sqrt{d1+e1 x} \sqrt{d2+e2 x}} dx \to \frac{1}{c^{m+1} \sqrt{-d_1 d_2}} \operatorname{Subst}\left[\int (a+b x)^n (c f+g \operatorname{Cosh}[x])^m dx, x, \operatorname{ArcCosh}[c x]\right]$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  1/(c^(m+1)*Sqrt[-d1*d2])*Subst[Int[(a+b*x)^n*(c*f+g*Cosh[x])^m,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && GtQ[d1,0] && LtQ[d2,0] && (GtQ[m,0] || IGtQ[n,0])
```

2: $\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land p+\frac{1}{2} \in \mathbb{Z}^- \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $e1 == c\, d1 \land e2 == -c\, d2 \land m \in \mathbb{Z} \land p+\frac{1}{2} \in \mathbb{Z}^- \land d1>0 \land d2<0 \land n \in \mathbb{Z}^+$, then

$$\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\int \frac{(a+b \operatorname{ArcCosh}[c x])^n}{\sqrt{d1+e1 x}\,\sqrt{d2+e2 x}} \operatorname{ExpandIntegrand}\left[(f+g x)^m (d1+e1 x)^{p+1/2} (d2+e2 x)^{p+1/2}, x\right] dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),(f+g*x)^m*(d1+e1*x)^(p+1/2)*(d2+e2*x)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && ILtQ[p+1/2,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0]
```

2: $\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $e1 == c\, d1 \wedge e2 == -c\, d2 \wedge m \in \mathbb{Z} \wedge p-\frac{1}{2} \in \mathbb{Z} \wedge \neg (d1>0 \wedge d2<0)$

Derivation: Piecewise constant extraction

Basis: If $e1 == c\, d1 \wedge e2 == -c\, d2$, then $\partial_x \frac{(d1+e1 x)^p (d2+e2 x)^p}{(-1+c x)^p (1+c x)^p} == 0$

Rule: If $e1 == c\, d1 \wedge e2 == -c\, d2 \wedge m \in \mathbb{Z} \wedge p-\frac{1}{2} \in \mathbb{Z} \wedge \neg (d1>0 \wedge d2<0)$, then

$$\int (f+g x)^m (d1+e1 x)^p (d2+e2 x)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$
$$\frac{(-d1\, d2)^{\operatorname{IntPart}[p]} (d1+e1 x)^{\operatorname{FracPart}[p]} (d2+e2 x)^{\operatorname{FracPart}[p]}}{(-1+c x)^{\operatorname{FracPart}[p]} (1+c x)^{\operatorname{FracPart}[p]}} \int (f+g x)^m (-1+c x)^p (1+c x)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/((-1+c*x)^FracPart[p]*(1+c*x)^FracPart[p])*
    Int[(f+g*x)^m*(-1+c*x)^p*(1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[m] && IntegerQ[p-1/2] && Not[GtQ[d1,0] && LtQ[d2,0]]
```

2. $\int \text{Log}\left[h\,(f+g\,x)^m\right]\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $e1 == c\,d1\,\land\,e2 == -c\,d2\,\land\,p-\frac{1}{2}\in\mathbb{Z}$

1. $\int \text{Log}\left[h\,(f+g\,x)^m\right]\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $e1 == c\,d1\,\land\,e2 == -c\,d2\,\land\,p-\frac{1}{2}\in\mathbb{Z}\,\land\,d1>0\,\land\,d2<0$

1: $\int \frac{\text{Log}\left[h\,(f+g\,x)^m\right]\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d1+e1\,x}\,\sqrt{d2+e2\,x}}\,dx$ when $e1 == c\,d1\,\land\,e2 == -c\,d2\,\land\,d1>0\,\land\,d2<0\,\land\,n\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: If $e1 == c\,d1\,\land\,e2 == -c\,d2\,\land\,d1>0\,\land\,d2<0$, then $\frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d1+e1\,x}\,\sqrt{d2+e2\,x}} == \partial_x\frac{(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{b\,c\,\sqrt{-d1\,d2}\,(n+1)}$

Basis: $\partial_x\,\text{Log}\left[h\,(f+g\,x)^m\right] == \frac{g\,m}{f+g\,x}$

Note: If $n\in\mathbb{Z}^+$, then $\frac{(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{f+g\,x}$ is integrable in closed-form.

Rule: If $e1 == c\,d1\,\land\,e2 == -c\,d2\,\land\,d1>0\,\land\,d2<0\,\land\,n\in\mathbb{Z}^+$, then

$$\int \frac{\text{Log}\left[h\,(f+g\,x)^m\right]\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d1+e1\,x}\,\sqrt{d2+e2\,x}}\,dx \longrightarrow \frac{\text{Log}\left[h\,(f+g\,x)^m\right]\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{b\,c\,\sqrt{-d1\,d2}\,(n+1)}-\frac{g\,m}{b\,c\,\sqrt{-d1\,d2}\,(n+1)}\int\frac{(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{f+g\,x}\,dx$$

Program code:

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  Log[h*(f+g*x)^m]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*Sqrt[-d1*d2]*(n+1)) -
  g*m/(b*c*Sqrt[-d1*d2]*(n+1))*Int[(a+b*ArcCosh[c*x])^(n+1)/(f+g*x),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,h,m},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[n,0]
```

2: $\int \text{Log}\left[h\,(f+g\,x)^m\right]\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $e1 == c\,d1\,\land\,e2 == -c\,d2\,\land\,p-\frac{1}{2}\in\mathbb{Z}\,\land\,\neg\,(d1>0\,\land\,d2<0)$

Derivation: Piecewise constant extraction

Basis: If $e1 == c\,d1\,\land\,e2 == -c\,d2$, then $\partial_x \frac{(d1+e1\,x)^p\,(d2+e2\,x)^p}{(-1+c\,x)^p\,(1+c\,x)^p} == 0$

–

Rule: If $e1 == c\,d1\,\land\,e2 == -c\,d2\,\land\,p-\frac{1}{2}\in\mathbb{Z}\,\land\,\neg\,(d1>0\,\land\,d2<0)$, then

$$\int \text{Log}\left[h\,(f+g\,x)^m\right]\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\,\longrightarrow$$

$$\frac{(-d1\,d2)^{\text{IntPart}[p]}\,(d1+e1\,x)^{\text{FracPart}[p]}\,(d2+e2\,x)^{\text{FracPart}[p]}}{(-1+c\,x)^{\text{FracPart}[p]}\,(1+c\,x)^{\text{FracPart}[p]}}\int \text{Log}\left[h\,(f+g\,x)^m\right]\,(-1+c\,x)^p\,(1+c\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$$

–

Program code:

```
Int[Log[h_.*(f_.+g_.*x_)^m_.]*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (-d1*d2)^IntPart[p]*(d1+e1*x)^FracPart[p]*(d2+e2*x)^FracPart[p]/((-1+c*x)^FracPart[p]*(1+c*x)^FracPart[p])*
    Int[Log[h*(f+g*x)^m]*(-1+c*x)^p*(1+c*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g,h,m,n},x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2] && Not[GtQ[d1,0] && LtQ[d2,0]]
```

6. $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$

1: $\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])\,dx$ when $m+\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Integration by parts

–

Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x]) == \frac{b\,c}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}$

Rule: If $m+\frac{1}{2}\in\mathbb{Z}^-$, let $u\to\int (d+e\,x)^m\,(f+g\,x)^m\,dx$, then

$$\int (d+e\,x)^m\,(f+g\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcCosh}[c\,x]) - b\,c\int \frac{u}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x)^m*(f+g*x)^m,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[Dist[1/(Sqrt[-1+c*x]*Sqrt[1+c*x]),u,x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && ILtQ[m+1/2,0]
```

2: $\int (d+e x)^m (f+g x)^m (a+b \text{ArcCosh}[c x])^n \, dx$ when $m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z}$, then

$$\int (d+e x)^m (f+g x)^m (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow \int (a+b \text{ArcCosh}[c x])^n \, \text{ExpandIntegrand}\left[(d+e x)^m (f+g x)^m, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_)^m_.*(f_+g_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n,(d+e*x)^m*(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && IntegerQ[m]
```

7: $\int u\,(a+b\,\text{ArcCosh}[c\,x])\,dx$ when $\int u\,dx$ is free of inverse functions

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x]) == \frac{b\,c}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}$

Rule: Let $v \to \int u\,dx$, if v is free of inverse functions, then

$$\int u\,(a+b\,\text{ArcCosh}[c\,x])\,dx \longrightarrow v\,(a+b\,\text{ArcCosh}[c\,x]) - b\,c\int\frac{v}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\,dx$$

$$\longrightarrow v\,(a+b\,\text{ArcCosh}[c\,x]) - \frac{b\,c\,\sqrt{1-c^2\,x^2}}{\sqrt{-1+c\,x}\,\sqrt{1+c\,x}}\int\frac{v}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[u_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{v=IntHide[u,x]},
  Dist[a+b*ArcCosh[c*x],v,x] -
  b*c*Sqrt[1-c^2*x^2]/(Sqrt[-1+c*x]*Sqrt[1+c*x])*Int[SimplifyIntegrand[v/Sqrt[1-c^2*x^2],x],x] /;
 InverseFunctionFreeQ[v,x]] /;
FreeQ[{a,b,c},x]
```

8. $\int P_x F\left[(d1+e1\,x)^p\,(d2+e2\,x)^p\right]\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $e1 == c\,d1 \land e2 == -c\,d2 \land p-\frac{1}{2}\in\mathbb{Z}$

1: $\int P_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $e1 == c\,d1 \land e2 == -c\,d2 \land p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

–

Rule: If $e1 == c\,d1 \land e2 == -c\,d2 \land p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int P_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx \longrightarrow \int \text{ExpandIntegrand}\left[P_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n,\,x\right]dx$$

–

Program code:

```
Int[Px_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && PolyQ[Px,x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2]
```

2: $\int P_x \left(f+g\,(d1+e1\,x)^p\,(d2+e2\,x)^p\right)^m (a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $e1==c\,d1\land e2==-c\,d2\land p+\frac{1}{2}\in\mathbb{Z}^+\land(m\mid n)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $e1==c\,d1\land e2==-c\,d2\land p+\frac{1}{2}\in\mathbb{Z}^+\land(m\mid n)\in\mathbb{Z}$, then

$$\int P_x \left(f+g\,(d1+e1\,x)^p\,(d2+e2\,x)^p\right)^m (a+b\,\text{ArcCosh}[c\,x])^n\,dx\longrightarrow\int \text{ExpandIntegrand}\left[P_x \left(f+g\,(d1+e1\,x)^p\,(d2+e2\,x)^p\right)^m (a+b\,\text{ArcCosh}[c\,x])^n,\,x\right]dx$$

Program code:

```
Int[Px_.*(f_+g_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Px*(f+g*(d1+e1*x)^p*(d2+e2*x)^p)^m*(a+b*ArcCosh[c*x])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,g},x] && PolyQ[Px,x] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IGtQ[p+1/2,0] && IntegersQ[m,n]
```

9. $\int RF_x\, u\, (a + b\, \text{ArcCosh}[c\, x])^n\, dx$ when $n \in \mathbb{Z}^+$

1. $\int RF_x\, (a + b\, \text{ArcCosh}[c\, x])^n\, dx$ when $n \in \mathbb{Z}^+$

1: $\int RF_x\, \text{ArcCosh}[c\, x]^n\, dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int RF_x\, \text{ArcCosh}[c\, x]^n\, dx \longrightarrow \int \text{ArcCosh}[c\, x]^n\, \text{ExpandIntegrand}[RF_x,\, x]\, dx$$

Program code:

```
Int[RFx_*ArcCosh[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[ArcCosh[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[c,x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2: $\int RF_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int RF_x\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx \longrightarrow \int \text{ExpandIntegrand}\left[RF_x\,(a+b\,\text{ArcCosh}[c\,x])^n,\ x\right] dx$$

Program code:

```
Int[RFx_*(a_+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[RFx*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2. $\int RF_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\wedge e1==c\,d1\wedge e2==-c\,d2\wedge p-\frac{1}{2}\in\mathbb{Z}$

1: $\int RF_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,\text{ArcCosh}[c\,x]^n\,dx$ when $n\in\mathbb{Z}^+\wedge e1==c\,d1\wedge e2==-c\,d2\wedge p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+\wedge e1==c\,d1\wedge e2==-c\,d2\wedge p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int RF_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,\text{ArcCosh}[c\,x]^n\,dx \longrightarrow \int (d1+e1\,x)^p\,(d2+e2\,x)^p\,\text{ArcCosh}[c\,x]^n\,\text{ExpandIntegrand}[RF_x,\ x]\,dx$$

Program code:

```
Int[RFx_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*ArcCosh[c_.*x_]^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(d1+e1*x)^p*(d2+e2*x)^p*ArcCosh[c*x]^n,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{c,d1,e1,d2,e2},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2]
```

2: $\int RF_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\land e1==c\,d1\land e2==-c\,d2\land p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+\land e1==c\,d1\land e2==-c\,d2\land p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int RF_x\,(d1+e1\,x)^p\,(d2+e2\,x)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\to\int(d1+e1\,x)^p\,(d2+e2\,x)^p\,\text{ExpandIntegrand}\left[RF_x\,(a+b\,\text{ArcCosh}[c\,x])^n,\,x\right]dx$$

Program code:

```
Int[RFx_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(d1+e1*x)^p*(d2+e2*x)^p,RFx*(a+b*ArcCosh[c*x])^n,x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && EqQ[e1-c*d1,0] && EqQ[e2+c*d2,0] && IntegerQ[p-1/2]
```

U: $\int u\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$

Rule:

$$\int u\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\to\int u\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,n},x]
```